RRI_Preferred_steering

* dsauersanjose@aol.com 12/17/08
* www.idea2ic.com

*



*

```
.OPTIONS  method=gear  GMIN=1e-18

VIN       INP     0       SIN(6   6       1K )
QP1       VP1     INP     VPNP    PNPV    1.04
QP2       VP2     OUT     VPNP    PNPV    1
QP3       VPNP    VP      VCC     PNPV    1
QP4       VP      VP      VCC     PNPV    1
RP1       VP1     0        5K
RP2       VP2     0        5K

QN1       VN1     INP     VNPN    NPNV    1.02
QN2       VN2     OUT     VNPN    NPNV    1
QN3       VNPN    VN      0       NPNV    1
```

```
QN4        VN     VN       0        NPNV    1
RN1        VCC    VN1      5K
RN2        VCC    VN2      5K

VCC        VCC     0       12
VREF       VREF    0       6
IPNP       VP5B    0       20u
RS         VP      VP5B    10k
QP5        VN      VP5B    VPNP      PNPV    1

BDITA      VIOUT   0      I = ( V(VP2) - V(VP1) + V(VN2) - V(VN1)  )/5000
BOTA       OUT     0      I = -1*( V(VIOUT) - V(VREF) )/5000
CCOMP      OUT     VIOUT  3p

.model       NPNV npn BF=150
.model       PNPV pnp BF=150
.model       PNPL pnp BF=5

.control
run
set         pensize = 2
dc          vin  0    12     100m
plot        v(out)-v(inp) title  OffsetVoltageVsCommonMode
plot        v(vp1)   v(vp2)  v(vcc)-v(vn1)  v(vcc)-v(vn2) title  TailCurrentsVsCommonMode
plot        v(vpnp)-v(vp5b) title  PnpEmitterBase
.endc
.end
```

*** ====================END==================================**

The preferred input stage is just a LM324 input stage that can swing to the positive rail. Most customers are going to be using a RRIO like a LM324, but they like the extra common mode range.



This means that there will only be one transition point where the offset voltage of the pnp's gets replaced by that of the npn's. This means some kind of current steering circuit needs to be build.



Transistor QP5 can do the job all by itself. When QP1 and QP2 are low, QP5 and QN1->QN4 are all off and all current flow in QP1 and QP2. At high supply the exact opposite is true.



The only problem is the emitter base voltage of QP5 which can get close to full supply.



That is not a problem if lateral pnps are used. The problem with that is that lateral pnps have at least a factor of ten less beta and are close to a thousand times slower that vertical pnps.

If vertical pnps are used, their emitter base voltages need to be protected. Applying a reverse voltage greater than a few volts is destructive.